
Elixir LangChain enables Elixir applications to integrate AI services and self-hosted models into an application.
Currently supported AI services:
	OpenAI ChatGPT
	OpenAI DALL-e 2 - image generation
	Anthropic Claude
	Google Gemini
	Google Vertex AI (Google's enterprise offering)
	Ollama
	Mistral
	Bumblebee self-hosted models - including Llama, Mistral and Zephyr
	LMStudio via their OpenAI compatibility API

LangChain is short for Language Chain. An LLM, or Large Language Model, is the "Language" part. This library makes it easier for Elixir applications to "chain" or connect different processes, integrations, libraries, services, or functionality together with an LLM.
LangChain is a framework for developing applications powered by language models. It enables applications that are:
	Data-aware: connect a language model to other sources of data
	Agentic: allow a language model to interact with its environment

The main value props of LangChain are:
	Components: abstractions for working with language models, along with a collection of implementations for each abstraction. Components are modular and easy-to-use, whether you are using the rest of the LangChain framework or not
	Off-the-shelf chains: a structured assembly of components for accomplishing specific higher-level tasks

Off-the-shelf chains make it easy to get started. For more complex applications and nuanced use-cases, components make it easy to customize existing chains or build new ones.

  
    
  
  What is this?


Large Language Models (LLMs) are emerging as a transformative technology, enabling developers to build applications that they previously could not. But using these LLMs in isolation is often not enough to create a truly powerful app - the real power comes when you can combine them with other sources of computation or knowledge.
This library is aimed at assisting in the development of those types of applications.

  
    
  
  Documentation


The online documentation can be found here.

  
    
  
  Demo


Check out the demo project that you can download and review.

  
    
  
  Relationship with JavaScript and Python LangChain


This library is written in Elixir and intended to be used with Elixir applications. The original libraries are LangChain JS/TS and LangChain Python.
The JavaScript and Python projects aim to integrate with each other as seamlessly as possible. The intended integration is so strong that that all objects (prompts, LLMs, chains, etc) are designed in a way where they can be serialized and shared between the two languages.
This Elixir version does not aim for parity with the JavaScript and Python libraries. Why not?
	JavaScript and Python are both Object Oriented languages. Elixir is Functional. We're not going to force a design that doesn't apply.
	The JS and Python versions started before conversational LLMs were standard. They put a lot of effort into preserving history (like a conversation) when the LLM didn't support it. We're not doing that here.

This library was heavily inspired by, and based on, the way the JavaScript library actually worked and interacted with an LLM.

  
    
  
  Installation


The package can be installed by adding langchain to your list of dependencies
in mix.exs:
def deps do
  [
    {:langchain, "0.2.0"}
  ]
end
The Release Candidate includes many additional features and some breaking changes.
def deps do
  [
    {:langchain, "0.3.0-rc.0"}
  ]
end

  
    
  
  Configuration


Currently, the library is written to use the Req library for making API calls.
You can configure an organization ID, and API key for OpenAI's API, but this library also works with other compatible APIs as well as other services and even local models running on Bumblebee.
config/runtime.exs:
config :langchain, openai_key: System.fetch_env!("OPENAI_API_KEY")
config :langchain, openai_org_id: System.fetch_env!("OPENAI_ORG_ID")
# OR
config :langchain, openai_key: "YOUR SECRET KEY"
config :langchain, openai_org_id: "YOUR_OPENAI_ORG_ID"

config :langchain, :anthropic_key, System.fetch_env!("ANTHROPIC_API_KEY")
It's possible to use a function or a tuple to resolve the secret:
config :langchain, openai_key: {MyApp.Secrets, :openai_api_key, []}
config :langchain, openai_org_id: {MyApp.Secrets, :openai_org_id, []}
# OR
config :langchain, openai_key: fn -> System.fetch_env!("OPENAI_API_KEY") end
config :langchain, openai_org_id: fn -> System.fetch_env!("OPENAI_ORG_ID") end
The API keys should be treated as secrets and not checked into your repository.
For fly.io, adding the secrets looks like this:
fly secrets set OPENAI_API_KEY=MyOpenAIApiKey
fly secrets set ANTHROPIC_API_KEY=MyAnthropicApiKey
A list of models to use:
	Anthropic Claude models
	Anthropic models on AWS Bedrock
	OpenAI models
	OpenAI models on Azure
	Gemini AI models


  
    
  
  Prompt caching


ChatGPT and Claude both offer prefix-based prompt caching, which can offer cost and performance benefits for longer prompts. Gemini offers context caching, which is similar.
	ChatGPT's prompt caching is automatic for prompts longer than 1024 tokens, caching the longest common prefix.
	Claude's prompt caching is not automatic. It's prefixing processes tools, system, and then messages, in that order, up to and including the block designated with {"cache_control": {"type": "ephemeral"}} . See LangChain.ChatModels.ChatAnthropicTest and for an example.
	Gemini's context caching requires a seperate call which is not supported by Langchain.


  
    
  
  Usage


The central module in this library is LangChain.Chains.LLMChain. Most other pieces are either inputs to this, or structures used by it. For understanding how to use the library, start there.

  
    
  
  Exposing a custom Elixir function to ChatGPT


A really powerful feature of LangChain is making it easy to integrate an LLM into your application and expose features, data, and functionality from your application to the LLM.
[image: Diagram showing LLM integration to application logic and data through a LangChain.Function]A LangChain.Function bridges the gap between the LLM and our application code. We choose what to expose and using context, we can ensure any actions are limited to what the user has permission to do and access.
For an interactive example, refer to the project Livebook notebook "LangChain: Executing Custom Elixir Functions".
The following is an example of a function that receives parameter arguments.
alias LangChain.Function
alias LangChain.Message
alias LangChain.Chains.LLMChain
alias LangChain.ChatModels.ChatOpenAI
alias LangChain.Utils.ChainResult

# map of data we want to be passed as `context` to the function when
# executed.
custom_context = %{
  "user_id" => 123,
  "hairbrush" => "drawer",
  "dog" => "backyard",
  "sandwich" => "kitchen"
}

# a custom Elixir function made available to the LLM
custom_fn =
  Function.new!(%{
    name: "custom",
    description: "Returns the location of the requested element or item.",
    parameters_schema: %{
      type: "object",
      properties: %{
        thing: %{
          type: "string",
          description: "The thing whose location is being requested."
        }
      },
      required: ["thing"]
    },
    function: fn %{"thing" => thing} = _arguments, context ->
      # our context is a pretend item/location location map
      {:ok, context[thing]}
    end
  })

# create and run the chain
{:ok, updated_chain}} =
  LLMChain.new!(%{
    llm: ChatOpenAI.new!(),
    custom_context: custom_context,
    verbose: true
  })
  |> LLMChain.add_tools(custom_fn)
  |> LLMChain.add_message(Message.new_user!("Where is the hairbrush located?"))
  |> LLMChain.run(mode: :while_needs_response)

# print the LLM's answer
IO.puts(update |> ChainResult.to_string())
# => "The hairbrush is located in the drawer."

  
    
  
  Alternative OpenAI compatible APIs


There are several services or self-hosted applications that provide an OpenAI compatible API for ChatGPT-like behavior. To use a service like that, the endpoint of the ChatOpenAI struct can be pointed to an API compatible endpoint for chats.
For example, if a locally running service provided that feature, the following code could connect to the service:
{:ok, updated_chain} =
  LLMChain.new!(%{
    llm: ChatOpenAI.new!(%{endpoint: "http://localhost:1234/v1/chat/completions"}),
  })
  |> LLMChain.add_message(Message.new_user!("Hello!"))
  |> LLMChain.run()

  
    
  
  Bumblebee Chat Support


Bumblebee hosted chat models are supported. There is built-in support for Llama 2, Mistral, and Zephyr models.
Currently, function calling is only supported for llama 3.1 Json Tool calling for Llama 2, Mistral, and Zephyr is NOT supported.
There is an example notebook in the notebook folder.
ChatBumblebee.new!(%{
  serving: @serving_name,
  template_format: @template_format,
  receive_timeout: @receive_timeout,
  stream: true
})
The serving is the module name of the Nx.Serving that is hosting the model.
See the LangChain.ChatModels.ChatBumblebee documentation for more details.

  
    
  
  Testing


To run all the tests including the ones that perform live calls against the OpenAI API, use the following command:
mix test --include live_call
mix test --include live_open_ai
mix test --include live_ollama_ai
mix test --include live_anthropic
mix test test/tools/calculator_test.exs --include live_call
NOTE: This will use the configured API credentials which creates billable events.
Otherwise, running the following will only run local tests making no external API calls:
mix test
Executing a specific test, whether it is a live_call or not, will execute it creating a potentially billable event.
When doing local development on the LangChain library itself, rename the .envrc_template to .envrc and populate it with your private API values. This is only used when running live test when explicitly requested.
Use a tool like Direnv or Dotenv to load the API values into the ENV when using the library locally.

  v0.3.1 (2025-02-05)



  
    
  
  Added


	Include stacktrace context in messages for caught exceptions from LLM functions & function callbacks. (#241)


  
    
  
  Changes


	Support LMStudio when using ChatOpenAI (#243)
	Fix issue with OpenAI converting an assistant message to JSON when the AI is talking while making tool calls in the same message (#245)


  
    
  
  v0.3.0 (2025-01-22)


No more breaking changes.

  
    
  
  Upgrading from v0.2.0 - v0.3.0


There were several breaking changes made in the different Release Candidates. All changes were kept. Refer to the CHANGELOG documentation for rc.0, rc.1 and rc.2 for specific examples and coverage on needed code updates.

  
    
  
  Added


	LLAMA 3.1 JSON tool call support with Bumblebee (#198)
	Raw field to TokenUsage (#236) - this returns the raw LLM token usage information, giving access to LLM-specific data.
	Prompt caching support for Claude (#226)
	Support for Ollama keep_alive API parameter (#237)
	Support for o1 OpenAI model (#234)
	Bumblebee Phi-4 support (#233)


  
    
  
  Changed


	Apply chat template from callback (#231)


  
    
  
  v0.3.0-rc.2 (2025-01-08)



  
    
  
  Breaking Changes


How LLM callbacks are registered has changed. The callback function's arguments have also changed.
Specifically, this refers to the callbacks:
	on_llm_new_delta
	on_llm_new_message
	on_llm_ratelimit_info
	on_llm_token_usage

The callbacks are still supported, but how they are registered and the arguments passed to the linked functions has changed.
Previously, an LLM callback's first argument was the chat model, it is now the LLMChain that is running it.
A ChatModel still has the callbacks struct attribute, but it should be considered private.
Why the change
Having some callback functions registered on the chat model and some registered on the chain was confusing. What goes where? Why the difference?
This change moves them all to the same place, removing a source of confusion.
The primary reason for the change is that important information about the context of the callback event was not available to the callback function. Information stored in the chain's custom_context can be valuable and important, like a user's account ID, but it was not easily accessible in a callback like on_llm_token_usage where we might want to record the user's token usage linked to their account.
This important change passes the entire LLMChain through to the callback function, giving the function access to the custom_context. This makes the LLM (aka chat model) callback functions expect the same arguments as the other chain focused callback functions.
This both unifies how the callbacks operate and what data they have available, and it groups them all together.
Adapting to the change
A before example:
llm_events = %{
  # 1st argument was the chat model
  on_llm_new_delta: fn _chat_model, %MessageDelta{} = delta ->
    # ...
  end,
  on_llm_token_usage: fn _chat_model, usage_data ->
    # ...
  end
}

chain_events = %{
  on_message_processed: fn _chain, tool_msg ->
    # ...
  end
}

# LLM callback events were registered on the chat model
chat_model = ChatOpenAI.new!(%{stream: true, callbacks: [llm_events]})

{:ok, updated_chain} =
  %{
    llm: chat_model,
    custom_context: %{user_id: 123}
  }
  |> LLMChain.new!()
  |> LLMChain.add_message(Message.new_system!())
  |> LLMChain.add_message(Message.new_user!("Say hello!"))
  # Chain callback events were registered on the chain
  |> LLMChain.add_callback(chain_events)
  |> LLMChain.run()
This is updated to: (comments highlight changes)
# Events are all combined together
events = %{
  # 1st argument is now the LLMChain
  on_llm_new_delta: fn _chain, %MessageDelta{} = delta ->
    # ...
  end,
  on_llm_token_usage: fn %LLMChain{} = chain, usage_data ->
    # ... `chain.custom_context` is available
  end,
  on_message_processed: fn _chain, tool_msg ->
    # ...
  end
}

# callbacks removed from Chat Model setup
chat_model = ChatOpenAI.new!(%{stream: true})

{:ok, updated_chain} =
  %{
    llm: chat_model,
    custom_context: %{user_id: 123}
  }
  |> LLMChain.new!()
  |> LLMChain.add_message(Message.new_system!())
  |> LLMChain.add_message(Message.new_user!("Say hello!"))
  # All events are registered through `add_callback`
  |> LLMChain.add_callback(events)
  |> LLMChain.run()
If you still need access to the LLM in the callback functions, it's available in chain.llm.
The change is a breaking change, but should be fairly easy to update.
This consolidates how callback events work and them more powerful by exposing important information to the callback functions.
If you were using the LLMChain.add_llm_callback/2, the change is even easier:
From:
  %{
    llm: chat_model,
    custom_context: %{user_id: 123}
  }
  |> LLMChain.new!()
  # ...
  # LLM callback events could be added later this way
  |> LLMChain.add_llm_callback(llm_events)
  |> LLMChain.run()
To:
  %{
    llm: chat_model,
    custom_context: %{user_id: 123}
  }
  |> LLMChain.new!()
  # ...
  # Use the `add_callback` function instead
  |> LLMChain.add_callback(llm_events)
  |> LLMChain.run()
Details of the change
	Removal of the LangChain.ChatModels.LLMCallbacks module.
	The LLM-specific callbacks were migrated to LangChain.Chains.ChainCallbacks.
	Removal of LangChain.Chains.LLMChain.add_llm_callback/2
	LangChain.ChatModels.ChatOpenAI.new/1 and LangChain.ChatModels.ChatOpenAI.new!/1 no longer accept :callbacks on the chat model.
	Removal of LangChain.ChatModels.ChatModel.add_callback/2


  
    
  
  What else Changed


	add explicit message support in summarizer by @brainlid in https://github.com/brainlid/langchain/pull/220
	Change abacus to optional dep by @nallwhy in https://github.com/brainlid/langchain/pull/223
	Remove constraint of alternating user, assistant by @GenericJam in https://github.com/brainlid/langchain/pull/222
	Breaking change: consolidate LLM callback functions by @brainlid in https://github.com/brainlid/langchain/pull/228
	feat: Enable :inet6 for Req.new for Ollama by @mpope9 in https://github.com/brainlid/langchain/pull/227
	fix: enable verbose_deltas by @cristineguadelupe in https://github.com/brainlid/langchain/pull/197


  
    
  
  New Contributors


	@nallwhy made their first contribution in https://github.com/brainlid/langchain/pull/223
	@GenericJam made their first contribution in https://github.com/brainlid/langchain/pull/222
	@mpope9 made their first contribution in https://github.com/brainlid/langchain/pull/227


  
    
  
  v0.3.0-rc.1 (2024-12-15)



  
    
  
  Breaking Changes


	Change return of LLMChain.run/2 (#170)
	Revamped error handling and handles Anthropic's "overload_error" - (#194)

Change return of LLMChain.run/2 (#170)
Why the change
Before this change, an LLMChain's run function returned {:ok, updated_chain, last_message}.
When an assistant (ie LLM) issues a ToolCall and when run is in the mode :until_success or :while_need_response, the LLMChain will automatically execute the function and return the result as a new Message back to the LLM. This works great!
The problem comes when an application needs to keep track of all the messages being exchanged during a run operation. That can be done by using callbacks and sending and receiving messages, but that's far from ideal. It makes more sense to have access to that information directly after the run operation completes.
What this change does
This PR changes the returned type to {:ok, updated_chain}.
The last_message is available in updated_chain.last_message. This cleans up the return API.
This change also adds %LLMChain{exchanged_messages: exchanged_messages},or updated_chain.exchanged_messages which is a list of all the messages exchanged between the application and the LLM during the execution of the run function.
This breaks the return contract for the run function.
How to adapt to this change
To adapt to this, if the application isn't using the last_message in {:ok, updated_chain, _last_message}, then delete the third position in the tuple. Ex: {:ok, updated_chain}.
Access to the last_message is available on the updated_chain.
{:ok, updated_chain} =
  %{llm: model}
  |> LLMChain.new!()
  |> LLMChain.run()

last_message = updated_chain.last_message
NOTE: that the updated_chain now includes updated_chain.exchanged_messages which can also be used.
Revamped error handling and handles Anthropic's "overload_error" - (#194)
What you need to do:
Check your application code for how it is responding to and handling error responses.
If you want to keep the same previous behavior, the following code change will do that:
case LLMChain.run(chain) do
  {:ok, _updated_chain} ->
    :ok

  # return the error for display
  {:error, _updated_chain, %LangChainError{message: reason}} ->
    {:error, reason}
end
The change from:
{:error, _updated_chain, reason}
To:
{:error, _updated_chain, %LangChainError{message: reason}}
When possible, a type value may be set on the LangChainError, making it easier to handle some error types programmatically.

  
    
  
  Features


	Added ability to summarize LLM conversations (#216)
	Implemented initial support for fallbacks (#207)
	Added AWS Bedrock support for ChatAnthropic (#154)
	Added OpenAI's new structured output API (#180)
	Added support for examples to title chain (#191)
	Added tool_choice support for OpenAI and Anthropic (#142)
	Added support for passing safety settings to Google AI (#186)
	Added OpenAI project authentication (#166)


  
    
  
  Fixes


	Fixed specs and examples (#211)
	Fixed content-part encoding and decoding for Google API (#212)
	Fixed ChatOllamaAI streaming response (#162)
	Fixed streaming issue with Azure OpenAI Service (#158, #161)
	Fixed OpenAI stream decode issue (#156)
	Fixed typespec error on Message.new_user/1 (#151)
	Fixed duplicate tool call parameters (#174)


  
    
  
  Improvements


	Added error type support for Azure token rate limit exceeded
	Improved error handling (#194)
	Enhanced function execution failure response
	Added "processed_content" to ToolResult struct (#192)
	Implemented support for strict mode for tools (#173)
	Updated documentation for ChatOpenAI use on Azure
	Updated config documentation for API keys
	Updated README examples


  
    
  
  Azure & Google AI Updates


	Added Azure test for ChatOpenAI usage
	Added support for system instructions for Google AI (#182)
	Handle functions with no parameters for Google AI (#183)
	Handle missing token usage fields for Google AI (#184)
	Handle empty text parts from GoogleAI responses (#181)
	Handle all possible finishReasons for ChatGoogleAI (#188)


  
    
  
  Documentation


	Added LLM Model documentation for tool_choice
	Updated documentation using new functions
	Added custom functions notebook
	Improved documentation formatting (#145)
	Added links to models in the config section
	Updated getting started doc for callbacks


  
    
  
  v0.3.0-rc.0 (2024-06-05)


Added:
	LangChain.ChatModels.ChatGoogleAI which differed too significantly from LangChain.ChatModels.ChatGoogleAI.  What's up with that? I'm looking at you Google! 👀	Thanks for the contribution Raul Chedrese!


	New callback mechanism was introduced to ChatModels and LLMChain. It was inspired by the approach used in the TS/JS LangChain library.
	Ability to provide plug-like middleware functions for pre-processing an assistant response message. Most helpful when coupled with a new run mode called :until_success. The first built-in one is LangChain.MessageProcessors.JsonProcessor.
	LLMChain has an internally managed current_failure_count and a publicly managed max_retry_count.
	New run mode :until_success uses failure and retry counts to repeatedly run the chain when the LLMs responses fail a MessageProcessor.
	LangChain.MessageProcessors.JsonProcessor is capable of extracting JSON contents and converting it to an Elixir map using Jason. Parsing errors are returned to the LLM for it to try again.
	The attribute processed_content was added to a LangChain.Message. When a MessageProcessor is run on a received assistant message, the results of the processing are accumulated there. The original content remains unchanged for when it is sent back to the LLM and used when fixing or correcting it's generated content.
	Callback support for LLM ratelimit information returned in API response headers. These are currently implemented for Anthropic and OpenAI.
	Callback support for LLM token usage information returned when available.
	LangChain.ChatModels.ChatModel additions	Added add_callback/2 makes it easier to add a callback to an chat model.
	Added serialize_config/1 to serialize an LLM chat model configuration to a map that can be restored later.
	Added restore_from_map/1 to restore a configured LLM chat model from a database (for example).


	LangChain.Chain.LLMChain additions	New function add_callback/2 makes it easier to add a callback to an existing LLMChain.
	New function add_llm_callback/2 makes it easier to add a callback to a chain's LLM. This is particularly useful when an LLM model is restored from a database when loading a past conversation and wanting to preserve the original configuration.



Changed:
	LLMChain.run/2 error result now includes the failed chain up to the point of failure. This is helpful for debugging.
	ChatOpenAI and ChatAnthropic both support the new callbacks.
	Many smaller changes and contributions were made. This includes updates to the README for clarity,
	LangChain.Utils.fire_callback/3 was refactored into LangChain.Utils.fire_streamed_callback/2 where it is only used for processing deltas and uses the new callback mechanism.
	Notebooks were moved to the separate demo project
	LangChain.ChatModels.ChatGoogleAI's key :version was changed to :api_version to be more consistent with other models and allow for model serializers to use the :version key.


  
    
  
  Migrations Steps


The LLMChain.run/2 function changed. Migrating should be easy.
From:
chain
|> LLMChain.run(while_needs_response: true)
Is changed to:
chain
|> LLMChain.run(mode: :while_needs_response)
This change enabled adding the new mode :until_success, which is mutually exclusive with :while_needs_response.
Additionally, the error return value was changed to include the chain itself.
From:
{:error, reason} = LLMChain.run(chain)
Is changed to:
{:error, _updated_chain, reason} = LLMChain.run(chain)
You can disregard the updated chain if you don't need it.
Callback events work differently now. Previously, a single callback_fn was executed and the developer needed to pattern match on a %Message{} or %MessageDelta{}. Callbacks work differently now.
When creating an LLM chat model, we can optionally pass in a map of callbacks where the event name is linked to the function to execute.
From:
live_view_pid = self()

callback_fn = fn
  %MessageDelta{} = delta ->
    send(live_view_pid, {:received_delta, delta})

  %Message{} = message ->
    send(live_view_pid, {:received_message, message})
end

{:ok, _result_chain, last_message} =
  LLMChain.new!(%{llm: %ChatAnthropic{stream: false}})
  |> LLMChain.add_message(Message.new_user!("Say, 'Hi!'!"))
  |> LLMChain.run(callback_fn: callback_fn)
The equivalent code would look like this:
Is changed to:
live_view_pid = self()

handler = %{
  on_llm_new_delta: fn _model, delta ->
    send(live_view_pid, {:received_delta, delta})
  end,
  on_llm_new_message: fn _model, message ->
    send(live_view_pid, {:received_message, message})
  end
}

{:ok, _result_chain, last_message} =
  LLMChain.new!(%{llm: %ChatAnthropic{stream: false, callbacks: [handler]}})
  |> LLMChain.add_message(Message.new_user!("Say, 'Hi!'!"))
  |> LLMChain.run()
The Message and MessageDelta callbacks are now set on the model. The callbacks are more granular and new callbacks are supported on the LLMChain as well. This more flexible configuration allows for more callbacks to be added as we move forward.
Also of note, is that the callbacks are set as a list of handler maps. This means we can assign multiple sets of callbacks for different purposes and they all get executed.

  
    
  
  v0.2.0 (2024-04-30)


For LLMs that support it (verified with ChatGPT and Anthropic), a user message can now contain multiple ContentParts, making it "multi-modal". This means images and text can be combined into a single message allowing for interactions about the images to now be possible.
Added:
	LangChain.Message.ContentPart - used for User messages and multi-modal support. Google's AI assistant can return multiple parts as well.
	LangChain.Message.ToolCall - an assistant can request multiple tool calls in the same message.
	LangChain.Message.ToolResult - the system's answer to a ToolCall. It adds an is_error boolean flag. This an be helpful in the UI, but Anthropic specifically wants it.
	Add llama-3 chat template by @bowyern in https://github.com/brainlid/langchain/pull/102

Changed:
	The roles of :function and :function_call are removed. The equivalent of a function_call is expressed by an :assistant role making one or more ToolCall requests. The :function was the system's answer to a function call. This is now in the :tool role.
	Role :tool was added. A tool message contains one or more ToolResult messages.


  
    
  
  v0.1.10 (2024-03-07)


Changes
	Fix invalid default url for google ai by @pkrawat1 in https://github.com/brainlid/langchain/pull/82


  
    
  
  v0.1.9 (2024-02-29) - The Leap Release!


This adds support for Bumblebee as a Chat model, making it easy to have conversations with Llama 2, Mistral, and Zephyr LLMs.
See the documentation in LangChain.ChatModels.ChatBumblebee for getting started.
NOTE: That at this time, none of the models support the function ability, so that is not supported yet.
This release includes an experimental change for better support of streamed responses that are broken up over multiple messages from services like ChatGPT and others.
Other library dependencies requirements were relaxed, making it easier to support different versions of libraries like req and nx.
	Add mistral chat by @michalwarda in https://github.com/brainlid/langchain/pull/76
	handle receiving JSON data broken up over multiple messages by @brainlid in https://github.com/brainlid/langchain/pull/80
	Add initial support for Zephyr 7b Beta by @brainlid in https://github.com/brainlid/langchain/pull/41


  
    
  
  v0.1.8 (2024-02-16)


Breaking change: RoutingChain's required values changed. Previously, default_chain was assigned an %LLMChain{} to return when no more specific routes matched.
This was changed to be default_route. It now expects a %PromptRoute{} to be provided.
Here's how to make the change:
  selected_route =
    RoutingChain.new(%{
      llm: ChatOpenAI.new(%{model: "gpt-3.5-turbo", stream: false}),
      input_text: user_input_text,
      routes: routes,
      default_route: PromptRoute.new!(%{name: "DEFAULT", chain: fallback_chain})
    })
    |> RoutingChain.evaluate()
The default_chain was changed to default_route and now expects a PromptRoute to be provided. The above example includes a sample default route that includes an optional fallback_chain.
Previously, the returned value from RoutingChain.evaluate/1 was a selected_chain; it now returns the selected_route.
Why was this changed?
This was changed to make it easier to use a PromptChain when there isn't an associated %LLMChain{} for it. The application must just need the answer of which route was selected.
This includes the change of not requiring a %PromptChain{}'s description or chain field.
Other Changes
	Add support for Ollama open source models by @medoror in https://github.com/brainlid/langchain/pull/70
	Add clause to match call_response spec by @medoror in https://github.com/brainlid/langchain/pull/72
	Add max_tokens option for OpenAI calls by @matthusby in https://github.com/brainlid/langchain/pull/73


  
    
  
  v0.1.7 (2024-01-18)


	Improvements for more intelligent agents - https://github.com/brainlid/langchain/pull/61	adds LangChain.Chains.RoutingChain - first-pass LLM chain to select the best route to take given the user's initial prompt
	adds LangChain.Chains.TextToTitleChain - turn the user's prompt into a title for the conversation


	Removed the need for a function to send a message to the process for how to display the function being executed
	Updated dependencies
	Add support for Google AI / Gemini Pro model by @jadengis in https://github.com/brainlid/langchain/pull/59
	Built-in automatic retries when underlying Mint connection is closed in https://github.com/brainlid/langchain/pull/68


  
    
  
  v0.1.6 (2023-12-12)


	Fix for correct usage of new Req retry setting. PR #57


  
    
  
  v0.1.5 (2023-12-11)


	Upgraded Req to v0.4.8. It contains a needed retry fix for certain situations.
	Fix OpenAI returns "Unrecognized request argument supplied: api_key" PR #54


  
    
  
  v0.1.4 (2023-12-11)


	Merged PR #45 - https://github.com/brainlid/langchain/pull/45	Added LangChain.Utils.ChainResult for helper functions when working with LLMChain result values.


	Merged PR #46 - https://github.com/brainlid/langchain/pull/46	Add possibility to use api_key per chat invocation.


	Merged PR #51 - https://github.com/brainlid/langchain/pull/51	Update req 0.4.7
	Hopefully resolves issue where Finch connections would be closed and a now does a built-in retry.


	Merged PR #52 - https://github.com/brainlid/langchain/pull/52	Allow overriding OpenAI compatible API endpoint. Caller can pass an alternate endpoint.




  
    
  
  v0.1.3 (2023-12-01)


	Merged PR #43 - https://github.com/brainlid/langchain/pull/43	Add Finch retry strategy to OpenAI Chat API requests


	Merged PR #39 - https://github.com/brainlid/langchain/pull/39	Changed ENV key from OPENAI_KEY to OPENAI_API_KEY to be consistent with the OpenAI docs.


	Merged PR #36 - https://github.com/brainlid/langchain/pull/36	Support specifying the seed with OpenAI calls. Used in testing for more deterministic behavior.


	Merged PR #34 - https://github.com/brainlid/langchain/pull/34	Enable sending the json_response flag with OpenAI model requests.


	Created LangChain.FunctionParam to express JSONSchema-friendly data structures. Supports basic types, arrays, enums, objects, arrays of objects and nested objects.	Still allows for full control over JSONSchema by providing an override parameters_schema object to full self-describe it.




  
    
  
  v0.1.2 (2023-10-26)


	refactor(chat_open_ai): Harden do_process_response by @Cardosaum in https://github.com/brainlid/langchain/pull/21	Improve JSON error handling result from ChatGPT


	Update req to 0.4.4 by @medoror in https://github.com/brainlid/langchain/pull/25	Updated to Req 0.4.4




  
    
  
  v0.1.1 (2023-10-10)


Minor update release.
	added "update_custom_context" to LLMChain
	added support for setting the OpenAI-Organization header in requests
	fixed data extraction chain and improved the prompt
	make chatgpt response tests more robust


  
    
  
  v0.1.0 (2023-09-18)


Initial release when published to hex.pm.


  

    
Getting Started
    

Mix.install([
  {:langchain, "~> 0.3.0"},
  {:kino, "~> 0.12.0"}
])

  
    
  
  Section


After installing the dependency, let's look at the simplest example to get started.
This is interactively available as a Livebook notebook named notebooks/getting_started.livemd.

  
    
  
  Basic Example


Let's build the simplest full LLMChain example so we can see how to make a call to ChatGPT from our Elixir application.
NOTE: This assumes your OPENAI_KEY is already set as a secret for this notebook.
Application.put_env(:langchain, :openai_key, System.fetch_env!("LB_OPENAI_API_KEY"))
alias LangChain.Chains.LLMChain
alias LangChain.ChatModels.ChatOpenAI
alias LangChain.Message

{:ok, updated_chain} =
  %{llm: ChatOpenAI.new!(%{model: "gpt-4o"})}
  |> LLMChain.new!()
  |> LLMChain.add_message(Message.new_user!("Testing, testing!"))
  |> LLMChain.run()

updated_chain.last_message.content
Nice! We've just saw how easy it is to get access to ChatGPT from our Elixir application!
Let's build on that example and define some system context for our conversation.

  
    
  
  Adding a System Message


When working with ChatGPT and other LLMs, the conversation works as a series of messages. The first message is the system message. This defines the context for the conversation. Here we can give the LLM some direction and impose limits on what it should do.
Let's create a system message followed by a user message.
{:ok, updated_chain} =
  %{llm: ChatOpenAI.new!(%{model: "gpt-4"})}
  |> LLMChain.new!()
  |> LLMChain.add_messages([
    Message.new_system!(
      "You are an unhelpful assistant. Do not directly help or assist the user."
    ),
    Message.new_user!("What's the capital of the United States?")
  ])
  |> LLMChain.run()

updated_chain.last_message.content
Here's the answer it gave me when I ran it:
Why don't you try looking it up online? There's so much information readily available on the internet. You might even learn a few other interesting facts about the country.

What I love about this is we can see the power of the system message. It completely changed the way the LLM behaves by default.
Beyond the system message, we pass back a whole collection of messages as the conversation continues. The updated_chain is part of the return and includes the newly received response message from the LLM as assistant message.

  
    
  
  Streaming Responses


If we want to display the messages as they are returned in the teletype way LLMs can, then we want to stream the responses.
In this example, we'll output the responses as they are streamed back. That happens in a callback function that we provide.
The stream: true setting belongs to the %ChatOpenAI{} struct that setups up our configuration. We also pass in the callbacks with the llm to fire the on_llm_new_delta. We can pass in the callbacks to the chain as well to fire the on_message_processed callback after the chain assembles the deltas and processes the finished message.
alias LangChain.MessageDelta

handler = %{
  on_llm_new_delta: fn _model, %MessageDelta{} = data ->
    # we received a piece of data
    IO.write(data.content)
  end,
  on_message_processed: fn _chain, %Message{} = data ->
    # the message was assmebled and is processed
    IO.puts("")
    IO.puts("")
    IO.inspect(data.content, label: "COMPLETED MESSAGE")
  end
}

{:ok, updated_chain} =
  %{
    # llm config for streaming and the deltas callback
    llm: ChatOpenAI.new!(%{model: "gpt-4o", stream: true})
  }
  |> LLMChain.new!()
  |> LLMChain.add_messages([
    Message.new_system!("You are a helpful assistant."),
    Message.new_user!("Write a haiku about the capital of the United States")
  ])
  # register the callbacks
  |> LLMChain.add_callback(handler)
  |> LLMChain.run()

updated_chain.last_message.content
# streamed
# ==> Washington D.C. stands,
# ... Monuments reflect history,
# ... Power's heart expands.

# ==> COMPLETED MESSAGE: "Washington D.C. stands,\nMonuments reflect history,\nPower's heart expands."
As the delta messages are received, the on_llm_new_delta callback function fires and the received data is written out to the console.
Finally, once the full message is received, the chain's on_message_processed callback fires and the completed message is written out separately.

  
    
  
  Next Steps


With the basics covered, you're ready to start integrating an LLM into your Elixir application! Check out other notebooks for more specific examples and other ways to use it.


  

    
Executing Custom Elixir Functions
    

Mix.install([
  {:langchain, "~> 0.3.0"}
])

  
    
  
  What we're doing


This notebook shows how to use the Elixir LangChain library to expose an Elixir function as a tool that can be executed by an LLM like ChatGPT, Anthropic and others. The LangChain library wraps this all up making it easy and portable between different LLMs.

  
    
  
  The Elixir Function in our App


Let's define the Elixir function we want to expose to ChatGPT so we can see how it works.
In this example we'll create a get_user_info function that takes a user ID and returns the relevant user's information for the current user to a web app.
For simplicity, we're skipping an actual database and storing our fake records on the module.
defmodule MyApp do
  @pretend_db %{
    1 => %{user_id: 1, name: "Michael Johnson", account_type: :trial, favorite_animal: "Horse"},
    2 => %{user_id: 2, name: "Joan Jett", account_type: :member, favorite_animal: "Aardvark"}
  }

  def get_user_info(user_id) do
    @pretend_db[user_id]
  end
end
{:module, MyApp, <<70, 79, 82, 49, 0, 0, 7, ...>>, {:get_user_info, 1}}
It's a simple lookup using a user_id to find return a map of a user's data.
MyApp.get_user_info(1)
%{name: "Michael Johnson", user_id: 1, account_type: :trial, favorite_animal: "Horse"}
With our function ready, let's cover how we can give the LLM access to this.

  
    
  
  Exposing our Function to an LLM


With an Elixir function defined, we will wrap it in a LangChain Function structure so it can be easily shared with an LLM.
This is what that looks like:
alias LangChain.Function

function =
  Function.new!(%{
    name: "get_user_info",
    description: "Return JSON object of the current users's relevant information.",
    function: fn _args, %{user_id: user_id} = _context ->
      # This uses the user_id provided through the context to call our Elixir function.
      {:ok, Jason.encode!(MyApp.get_user_info(user_id))}
    end
  })
%LangChain.Function{
  name: "get_user_info",
  description: "Return JSON object of the current users's relevant information.",
  display_text: nil,
  strict: false,
  function: #Function<41.18682967/2 in :erl_eval.expr/6>,
  async: true,
  parameters_schema: nil,
  parameters: []
}
The function name we provide is how the LLM will execute the function if the LLM chooses to call it.
The description is for the LLM to know what the function can do so it can decide which function to call for which purpose.
The function argument is passed an annonymous function whose job it is to be the glue that bridges data coming from the LLM with context from our application before calling other functions from our application.
The LangChain.Function acts as a bridge between the LLM and our application. The Elixir function receives 2 arguments. The first is any arguments passed to the function by the LLM if we defined any as being required. In this example, the LLM doesn't provide any arguments. The second argument is an application context that we'll get to next.
The context is specific to our application and does not go through the LLM at all. Think of this as the current user logged into our Phoenix web application. We want the user's interaction with the LLM to be relevant and limited to only what the current user can see and do.
For returning the final result, since the LLM responses must be text, we convert the returned Map into JSON. Additionally, we return it inside an {:ok, json} tuple, letting the library know that the function call was successful.

  
    
  
  Setting up our LangChain API Key


We need to setup the LangChain library to connect with ChatGPT using our API key. In a real Elixir application, this would be done in the config/config.exs file using something like this:
config :langchain, :openai_key, fn -> System.fetch_env!("OPENAI_API_KEY") end
For the Livebook notebook, use the "Secrets" on the sidebar to create an OPENAI_API_KEY secret with you API key. That is accessible here using "LB_OPENAI_API_KEY".
Application.put_env(:langchain, :openai_key, System.fetch_env!("LB_OPENAI_API_KEY"))
:ok

  
    
  
  Defining our AI Assistant


We'll use the LangChain.Message struct to define the messages for what we want the LLM to do. Our system message instructs the LLM how to behave.
In this example, we want the assistant to generate Haiku poems about the current user's favorite animal. However, we only want it to work for users who are "members" and not "trial" users.
The instructions we're giving the LLM will require it to execute the function to get additional information. Yes, this is both a simple and contrived example. In a real system, we wouldn't even make the API call to the server for a "trial" user and we could pass along the additional information with the first request.
What we're demonstrating here is that the LLM can interact with our Elixir application, use multiple pieces of returned information to make business logic decisions and fullfil our system requests.
alias LangChain.Message

messages = [
  Message.new_system!(~s(You are a helpful haiku poem generating assistant.
    ONLY generate a haiku for users with an `account_type` of "member".
    If the user has an `account_type` of "trial", say you can't do it,
    but you would love to help them if they upgrade and become a member.)),
  Message.new_user!("The current user is requesting a Haiku poem about their favorite animal.")
]
[
  %LangChain.Message{
    content: "You are a helpful haiku poem generating assistant.\n    ONLY generate a haiku for users with an `account_type` of \"member\".\n    If the user has an `account_type` of \"trial\", say you can't do it,\n    but you would love to help them if they upgrade and become a member.",
    processed_content: nil,
    index: nil,
    status: :complete,
    role: :system,
    name: nil,
    tool_calls: [],
    tool_results: nil
  },
  %LangChain.Message{
    content: "The current user is requesting a Haiku poem about their favorite animal.",
    processed_content: nil,
    index: nil,
    status: :complete,
    role: :user,
    name: nil,
    tool_calls: [],
    tool_results: nil
  }
]

  
    
  
  Defining our AI Model


For this example, we're talking to OpenAI's ChatGPT service. Let's setup that model. At this point, we can also specify which version of ChatGPT we want to talk with.
For the kind of work we're asking it to do, GPT-4o is the latest model that has great support for tools.
alias LangChain.ChatModels.ChatOpenAI

chat_model = ChatOpenAI.new!(%{model: "gpt-4o", temperature: 1, stream: false})
%LangChain.ChatModels.ChatOpenAI{
  endpoint: "https://api.openai.com/v1/chat/completions",
  model: "gpt-4o",
  api_key: nil,
  temperature: 1.0,
  frequency_penalty: 0.0,
  reasoning_mode: false,
  reasoning_effort: "medium",
  receive_timeout: 60000,
  seed: nil,
  n: 1,
  json_response: false,
  json_schema: nil,
  stream: false,
  max_tokens: nil,
  stream_options: nil,
  tool_choice: nil,
  callbacks: [],
  user: nil
}

  
    
  
  Defining our Application's User Context


Here we'll define some special context that we want passed through to our LangChain.Function when it is executed.
In a real application, this might be session based user or account information. It's whatever is relevant to our application that changes how a function should operate or the data it should access.
context = %{user_id: 2}
%{user_id: 2}
After trying this with user_id: 2, a member who should have a Haiku generated for them, change it to user_id: 1 to see it be polietly denied.

  
    
  
  Making the API Call


We're ready to make the API call!
Notice the custom_context: context setting that is passed in when creating the LLMChain. That information is the application-specific context we want to be passed to our Function when executed.
Also, note the verbose: true setting. That causes a number of IO.inspect calls to be printed showing what's happening internally.
Additionally, the stream: false option says we want the result only when it's complete. This example isn't setup for receving a streaming response. We're keeping it simple!
alias LangChain.Chains.LLMChain

{:ok, updated_chain} =
  %{llm: chat_model, custom_context: context, verbose: true}
  |> LLMChain.new!()
  |> LLMChain.add_messages(messages)
  |> LLMChain.add_tools([function])
  # keep running the LLM chain against the LLM if needed to evaluate
  # function calls and provide a response.
  |> LLMChain.run(mode: :while_needs_response)

response = updated_chain.last_message
IO.write(response.content)
response.content
LLM: %LangChain.ChatModels.ChatOpenAI{
  endpoint: "https://api.openai.com/v1/chat/completions",
  model: "gpt-4o",
  api_key: nil,
  temperature: 1.0,
  frequency_penalty: 0.0,
  reasoning_mode: false,
  reasoning_effort: "medium",
  receive_timeout: 60000,
  seed: nil,
  n: 1,
  json_response: false,
  json_schema: nil,
  stream: false,
  max_tokens: nil,
  stream_options: nil,
  tool_choice: nil,
  callbacks: [],
  user: nil
}
MESSAGES: [
  %LangChain.Message{
    content: "You are a helpful haiku poem generating assistant.\n    ONLY generate a haiku for users with an `account_type` of \"member\".\n    If the user has an `account_type` of \"trial\", say you can't do it,\n    but you would love to help them if they upgrade and become a member.",
    processed_content: nil,
    index: nil,
    status: :complete,
    role: :system,
    name: nil,
    tool_calls: [],
    tool_results: nil
  },
  %LangChain.Message{
    content: "The current user is requesting a Haiku poem about their favorite animal.",
    processed_content: nil,
    index: nil,
    status: :complete,
    role: :user,
    name: nil,
    tool_calls: [],
    tool_results: nil
  }
]
TOOLS: [
  %LangChain.Function{
    name: "get_user_info",
    description: "Return JSON object of the current users's relevant information.",
    display_text: nil,
    strict: false,
    function: #Function<41.18682967/2 in :erl_eval.expr/6>,
    async: true,
    parameters_schema: nil,
    parameters: []
  }
]
SINGLE MESSAGE RESPONSE: %LangChain.Message{
  content: nil,
  processed_content: nil,
  index: 0,
  status: :complete,
  role: :assistant,
  name: nil,
  tool_calls: [
    %LangChain.Message.ToolCall{
      status: :complete,
      type: :function,
      call_id: "call_2puDjavxfbFQ7uDvrE9vafnZ",
      name: "get_user_info",
      arguments: %{},
      index: nil
    }
  ],
  tool_results: nil
}
MESSAGE PROCESSED: %LangChain.Message{
  content: nil,
  processed_content: nil,
  index: 0,
  status: :complete,
  role: :assistant,
  name: nil,
  tool_calls: [
    %LangChain.Message.ToolCall{
      status: :complete,
      type: :function,
      call_id: "call_2puDjavxfbFQ7uDvrE9vafnZ",
      name: "get_user_info",
      arguments: %{},
      index: nil
    }
  ],
  tool_results: nil
}
EXECUTING FUNCTION: "get_user_info"

09:10:30.474 [debug] Executing function "get_user_info"
FUNCTION RESULT: "{\"name\":\"Joan Jett\",\"user_id\":2,\"account_type\":\"member\",\"favorite_animal\":\"Aardvark\"}"
TOOL RESULTS: %LangChain.Message{
  content: nil,
  processed_content: nil,
  index: nil,
  status: :complete,
  role: :tool,
  name: nil,
  tool_calls: [],
  tool_results: [
    %LangChain.Message.ToolResult{
      type: :function,
      tool_call_id: "call_2puDjavxfbFQ7uDvrE9vafnZ",
      name: "get_user_info",
      content: "{\"name\":\"Joan Jett\",\"user_id\":2,\"account_type\":\"member\",\"favorite_animal\":\"Aardvark\"}",
      processed_content: nil,
      display_text: nil,
      is_error: false
    }
  ]
}
SINGLE MESSAGE RESPONSE: %LangChain.Message{
  content: "Long snout seeking ants,  \nNight wanderer of moonlight,  \nAardvark's gentle grace.",
  processed_content: nil,
  index: 0,
  status: :complete,
  role: :assistant,
  name: nil,
  tool_calls: [],
  tool_results: nil
}
MESSAGE PROCESSED: %LangChain.Message{
  content: "Long snout seeking ants,  \nNight wanderer of moonlight,  \nAardvark's gentle grace.",
  processed_content: nil,
  index: 0,
  status: :complete,
  role: :assistant,
  name: nil,
  tool_calls: [],
  tool_results: nil
}
Long snout seeking ants,  
Night wanderer of moonlight,  
Aardvark's gentle grace.
"Long snout seeking ants,  \nNight wanderer of moonlight,  \nAardvark's gentle grace."
TIP: Try changing the context to user_id: 1 now and see what happens when a different user context is provided.

  
    
  
  Discussion


After a successful call, we can see in the verbose logs that:
	the LLM requested to execute the tool which is our function
	LLMChain executed the Elixir function attached to the Function struct
	the response of our Elixir function passed through the anonymous function on Function and was re-submitted back to the LLM
	the LLM reacted to the result of our function call

This means it worked! We successfully let an LLM directly interact with our Elixir application!
With this, we could expose functions that allow the LLM to request additional information specific to the current user, or we could even define functions that allow the LLM to change things in the user's account on their behalf!
The rest is up to us.


  

    
Images: Generating context-specific descriptions
    

Mix.install([
  {:langchain, "~> 0.3.0"},
  {:kino, "~> 0.12.0"}
])

  
    
  
  Overview


For the AI, we will leverage both OpenAI's ChatGPT and Anthropic's Claude LLMs (Large Language Models) through the Elixir LangChain library to perform AI analysis on images. In particular, we want context-aware caption text and shorter context-aware image ALT text. "Context-aware" means that the description should relate to how the image is being used and not simply a description of the image. With these two versions of an image description, we can improve SEO (Search Engine Optimization) and improve the accessibility of our content as well.

  
    
  
  It starts with an image


Before we can interact with an LLM, we need an image to work with. There's an image input to make it easy to upload an image.
You can take a photo or upload an image to use. This notebook example was based on this image: unsplash.com/photos/two-woman-walking-under-bridge-DcNlJK7kLkk

  
    
  
  Scenario


In this example, Sarah is working on a feature article for an online magazine focused on "The Hidden Gems of Urban Street Art." Her article aims to showcase the vibrant and often overlooked artworks that adorn the nooks and crannies around Toronto Canada.
The article should also be accessible to readers who use assistive technology like screenreaders. There are a number of images for the article and we want to generate high quality ALT text or caption text for the images and we want it written with awareness of the article and the context for witch the text will be used.
We're developing a solution for this problem in our notebook, but our solution is able to be scaled up to operated on hundreds of images if needed.
Let's get started by getting the image into the notebook.
input = Kino.Input.image("image", format: :jpeg)
image = Kino.Input.read(input)
The LLMs we're working with expect the image to be supplied as base64 encoded text. Let's get that ready here:
image_data =
  image.file_ref
  |> Kino.Input.file_path()
  |> File.read!()
  |> Base.encode64()

:ok

  
    
  
  Elixir LangChain


To make the requests for AI help, we'll use the Elixir LangChain library. It makes it much easier to integrate AI into our Elixir applications.
With our image ready, we'll setup a request to ChatGPT and ask it to write a context aware description of the image.
NOTE: You must provide your own OPENAI_API_KEY in the Livebook secrets to do this.
Application.put_env(:langchain, :openai_key, System.fetch_env!("LB_OPENAI_API_KEY"))
First, let's setup the model for talking to ChatGPT. For simplicity, we've set it not stream the response back. We'll get the final analysis once it's complete.
NOTE: For ChatGPT, image support requires using the "gpt-4o" model at the time this was created.
alias LangChain.ChatModels.ChatOpenAI

openai_chat_model = ChatOpenAI.new!(%{model: "gpt-4o"})
Here we setup our messages. The user message contains multiple ContentParts. One is text for our prompt and the secone is the base64 encoded image data. We submit all of this data together in our request.
This is where we add context to our image description request. We'll assume that we have programmatic access to the images and that we also have access to some data about the image from an external system.
TIP: To get consitency across all the images in our set, we'll have better success by providing an example of the type of output we want.
NOTE: Make sure the :media option matches both the image and what is supported by the LLM you are connecting with.
alias LangChain.Message
alias LangChain.Message.ContentPart
alias LangChain.PromptTemplate

messages = [
  Message.new_system!("""
  You are an expert at providing an image description for assistive technology and SEO benefits.

  The image is included in an online article titled "The Hidden Gems of Urban Street Art."

  The article aims to showcase the vibrant and often overlooked artworks that adorn
  the nooks and crannies around the city of Toronto Canada.

  You generate text for two purposes:
  - an HTML img alt text
  - an HTML figure, figcaption text

  ## Alt text format
  Briefly describe the contents of the image where the context is focusing on the urban street art.
  Be concise and limit the description to 125 characters or less.

  Example alt text:
  > A vibrant phoenix graffiti with blazing orange, red, and gold colors on the side of a brick building in an urban setting.

  ## figcaption format
  Image caption descriptions should focus on the urban artwork, providing a description of the appearance,
  style, street address if available, and how it relates to the surroundings. Be concise.

  Example caption text:
  > A vibrant phoenix graffiti on a brick building at Queen St W and Spadina Ave. With wings outstretched, the mural's blazing oranges, reds, and golds contrast sharply against the red brick backdrop. Passersby pause to observe, integrating the artwork into the urban landscape.
  """),
  Message.new_user!([
    PromptTemplate.from_template!("""
    Provide the descriptions for the image. Incorporate relevant information from the following additional details if applicable:

    <%= @extra_image_info %>

    Output in the following format:

    ```json
    {
      "alt": "generated alt text",
      "caption": "generation caption text"
    }
    ```
    """),
    ContentPart.image!(image_data, media: :jpg, detail: "low")
  ])
]
Before we continue, notice that the System message provides the general context for what we are doing and what we want from the LLM.
The User message is made up of two parts:
	PromptTemplate: supports variable replacement tags using EEx templates. This allows us to easily customize the prompt for each image as we process through a whole batch. This turns into a ContentPart.
	ContentPart: Makes it easy for us to provide our image directly to the LLM.

We are providing the LLM with the context for the task, specific instructions about an image, and an image to analyze with a "vision" enabled model so it can finally perform the task for us.
Note the use of a PromptTemplate and <%= @image_data %> in our user Message. As we are processing a whole set of images, data from the system where we get the image is rendered into our prompt, helping to further customize the generated text from the LLM, making it far more specific and relevant.

  
    
  
  JSON Output


For each image we want two pieces of generated content. To make this easy on ourselves, we instruct the LLM to output the two pieces of information in a JSON object.
Specifically, we instruct it to output in the following format:
{
  "alt": "generated alt text",
  "caption": "generation caption text"
}
To make working with that output easier, we'll use a JsonProcessor for processing messages from the LLM. It has the added ability to return a JSON formatting error to the LLM in the case when it get's it wrong.
We'll see this next when we put it all together.

  
    
  
  Making the Request


Everything is ready to make the request!
	We have the image
	We setup which LLM we are connecting with
	We provide context in our prompt and instructions for the type of description we want

Now, we'll submit the request to the server and review the response. For this example, the "image_data_from_other_system" is a substitute for a database call or other lookup for the information we have on the image.
alias LangChain.Chains.LLMChain
alias LangChain.MessageProcessors.JsonProcessor

# This data comes from an external data source per image.
# When we `apply_prompt_templates` below, the data is rendered into the template.
image_data_from_other_system = "image of urban art mural on underpass at 507 King St E"

{:ok, updated_chain} =
  %{llm: openai_chat_model, verbose: true}
  |> LLMChain.new!()
  |> LLMChain.apply_prompt_templates(messages, %{extra_image_info: image_data_from_other_system})
  |> LLMChain.message_processors([JsonProcessor.new!(~r/```json(.*?)```/s)])
  |> LLMChain.run(mode: :until_success)

updated_chain.last_message.processed_content
Notice that when running the chain, we use the option mode: :until_success. Some LLMs are better are generating valid JSON than other LLMs. When we included the JsonProcesser, it parses the assistant's content, converting it into an Elixir map. The converted data is stored on the message.processed_content.
If the LLM fails to give us valid JSON, the JsonProcessor generates a :user message reporting the issue for the LLM. The :until_success mode will repeatedly make the round-trip requests to the LLM, giving it an opportunity to fix it. But don't worry, it won't run forever! The LLMChain's max_retry_count gives up after X failures, the default being 3.
Here's a sample of what was generated when I ran it:
%{
  "alt" => "Colorful mural of a face under a bridge at 507 King St E",
  "caption" => "A captivating face mural under the 507 King St E underpass, featuring vivid hues and expressive eyes. The art adds a pop of color to the urban landscape, drawing the attention of pedestrians as they pass by."
}
This demonstrates how adding context to an image description request can really become immediately useful. The same system prompt and message template can be used for the whole set of images and generate high quality descriptive text for images.
Will it be as good as a human? No, honestly it may not be. However, it can be done much faster, automated and performed on more images than a human could do. In many instances, that means we get good, context aware ALT text where we may otherwise have none because we don't have the people nor the time to manually create it.
The result is our visual content is more accessible to people using assistive technology like a screen reader.

  
    
  
  Anthropic (BONUS)


While we're all setup for it, if you have an Anthropic API key, then we'll submit the same request to Anthropic and see how that compares.
NOTE: You must provide your own ANTHROPIC_API_KEY in the Livebook secrets to do this.
Application.put_env(:langchain, :anthropic_key, System.fetch_env!("LB_ANTHROPIC_API_KEY"))
Let's setup our Anthropic chat model.
NOTE: Keep in mind that different versions of Claude will give different results. You can play with that to find a good cost/accuracy for your specific need.
alias LangChain.ChatModels.ChatAnthropic

anthropic_chat_model = ChatAnthropic.new!(%{model: "claude-3-5-sonnet-latest"})
Now we run the same messages through an identical LLMChain but passing in the Anthropic chat model.
alias LangChain.Chains.LLMChain
alias LangChain.MessageProcessors.JsonProcessor

# This data comes from an external data source per image.
# When we `apply_prompt_templates` below, the data is rendered into the template.
image_data_from_other_system = "image of urban art mural on underpass at 507 King St E"

{:ok, updated_chain} =
  %{llm: anthropic_chat_model, verbose: true}
  |> LLMChain.new!()
  |> LLMChain.apply_prompt_templates(messages, %{extra_image_info: image_data_from_other_system})
  |> LLMChain.message_processors([JsonProcessor.new!(~r/```json(.*?)```/s)])
  |> LLMChain.run(mode: :until_success)

updated_chain.last_message.processed_content
Nice! The Elixir LangChain library abstracted away the differences between the two services. With no code changes, we can make a similar request about the image from Anthropic's Claude LLM as well!
Here's what I got from it:
%{
  "alt" => "Colorful street art mural with a face in vibrant colors on an underpass pillar in an urban setting.",
  "caption" => "A striking urban art mural adorns an underpass pillar at 507 King St E. The artwork features a mesmerizing face composed of vivid, rainbow-like hues and intricate patterns. Framed by the industrial yellow beams above, the mural transforms the concrete structure into a captivating focal point for passersby."
}
We would want to run multiple tests on a small sampling of images and tweak our prompt until we are happy with the result. Then, we can process full batch and save our work as a template for future projects as well.

Decodes AWS messages in the application/vnd.amazon.eventstream content-type.
Ignores the headers because on Bedrock it's the same content type, event type & message type headers in every message.

LangChain.ChatModels.ChatAnthropic 
    



      
Module for interacting with Anthropic models.
Parses and validates inputs for making requests to Anthropic's messages API.
Converts responses into more specialized LangChain data structures.

  
    
  
  Callbacks


See the set of available callback: LangChain.ChatModels.LLMCallbacks

  
    
  
  Rate Limit API Response Headers


Anthropic returns rate limit information in the response headers. Those can be
accessed using an LLM callback like this:
handler = %{
  on_llm_ratelimit_info: fn _chain, headers ->
    IO.inspect(headers)
  end
}

%{llm: ChatAnthropic.new!(%{model: "..."})}
|> LLMChain.new!()
# ... add messages ...
|> LLMChain.add_callback(handler)
|> LLMChain.run()
When a request is received, something similar to the following will be output
to the console.
%{
  "anthropic-ratelimit-requests-limit" => ["50"],
  "anthropic-ratelimit-requests-remaining" => ["49"],
  "anthropic-ratelimit-requests-reset" => ["2024-06-08T04:28:30Z"],
  "anthropic-ratelimit-tokens-limit" => ["50000"],
  "anthropic-ratelimit-tokens-remaining" => ["50000"],
  "anthropic-ratelimit-tokens-reset" => ["2024-06-08T04:28:30Z"],
  "request-id" => ["req_1234"]
}

  
    
  
  Tool Choice


Anthropic supports forcing a tool to be used.
	https://docs.anthropic.com/en/docs/build-with-claude/tool-use#forcing-tool-use

This is supported through the tool_choice options. It takes a plain Elixir map to provide the configuration.
By default, the LLM will choose a tool call if a tool is available and it determines it is needed. That's the "auto" mode.

  
    
  
  Example


For the LLM's response to make a tool call of the "get_weather" function.
ChatAnthropic.new(%{
  model: "...",
  tool_choice: %{"type" => "tool", "name" => "get_weather"}
})

  
    
  
  AWS Bedrock Support


Anthropic Claude is supported in AWS Bedrock.
To configure ChatAnthropic for use on AWS Bedrock:
	Request Model Access to get access to the Anthropic models you intend to use.

	Using your AWS Console, create an Access Key for your application.

	Set the key values in your AWS_ACCESS_KEY_ID and AWS_SECRET_ACCESS_KEY ENVs.

	Get the Model ID for the model you intend to use. Base Models

	Refer to LangChain.Utils.BedrockConfig for setting up the Bedrock authentication credentials for your environment.

	Setup your ChatAnthropic similar to the following:
 alias LangChain.ChatModels.ChatAnthropic
 ChatAnthropic.new!(%{
   model: "anthropic.claude-3-5-sonnet-20241022-v2:0",
   bedrock: BedrockConfig.from_application_env!()
 })

Represents a chat model hosted by Bumblebee and accessed through an
Nx.Serving.
Many types of models can be hosted through Bumblebee, so this attempts to
represent the most common features and provide a single implementation where
possible.
For streaming responses, the Bumblebee serving must be configured with
stream: true and should include stream_done: true as well.
Example:
Bumblebee.Text.generation(model_info, tokenizer, generation_config,
  # ...
  stream: true,
  stream_done: true
)
This supports a non streaming response as well, in which case, a completed
LangChain.Message is returned at the completion.
The stream_done option sends a final message to let us know when the stream
is complete and includes some token information.
The chat model can be created like this and provided to an LLMChain:
ChatBumblebee.new!(%{
  serving: @serving_name,
  template_format: @template_format,
  receive_timeout: @receive_timeout,
  stream: true
})
The serving is the module name of the Nx.Serving that is hosting the
model.
The following are the supported values for template_format. These are
provided by LangChain.Utils.ChatTemplates.
Chat models are trained against specific content formats for the messages.
Some models have no special concept of a system message. See the
LangChain.Utils.ChatTemplates documentation for specific format examples.
Using the wrong format with a model may result in poor performance or
hallucinations. It will not result in an error.

  
    
  
  Full example of chat through Bumblebee


Here's a full example of having a streaming conversation with Llama 2 through
Bumblebee.
defmodule MyApp.BumblebeeChat do
  @doc false
  alias LangChain.Message
  alias LangChain.ChatModels.ChatBumblebee
  alias LangChain.Chains.LLMChain

  def run_chat do
    # Used when streaming responses. The function fires as data is received.
    callback_fn = fn
      %LangChain.MessageDelta{} = delta ->
        # write to the console as the response is streamed back
        IO.write(delta.content)

      %LangChain.Message{} = message ->
        # inspect the fully finished message that was assembled from all the deltas
        IO.inspect(message, label: "FULLY ASSEMBLED MESSAGE")
    end

    # create and run the chain
    {:ok, _updated_chain, %Message{} = message} =
      LLMChain.new!(%{
        llm:
          ChatBumblebee.new!(%{
            serving: Llama2ChatModel,
            template_format: :llama_2,
            stream: true
          }),
        verbose: true
      })
      |> LLMChain.add_message(Message.new_system!("You are a helpful assistant."))
      |> LLMChain.add_message(Message.new_user!("What is the capital of Taiwan? And share up to 5 interesting facts about the city."))
      |> LLMChain.run(callback_fn: callback_fn)

    # print the LLM's fully assembled answer
    IO.puts("\n\n")
    IO.puts(message.content)
    IO.puts("\n\n")
  end
end
Then run the code in IEx:
  recompile; MyApp.BumblebeeChat.run_chat

LangChain.ChatModels.ChatGoogleAI 
    



      
Parses and validates inputs for making a request for the Google AI  Chat API.
Converts response into more specialized LangChain data structures.
NOTE: The GoogleAI service is unique in how it reports TokenUsage
information. So far, it's the only API that returns TokenUsage for each
returned delta, where the generated token count is incremented with one. Other
services return the total TokenUsage data at the end. This Chat model fires
the callback each time it is received.

LangChain.ChatModels.ChatOllamaAI 
    



      
Represents the Ollama AI Chat model
Parses and validates inputs for making a requests from the Ollama Chat API.
Converts responses into more specialized LangChain data structures.
The module's functionalities include:
	Initializing a new ChatOllamaAI struct with defaults or specific attributes.
	Validating and casting input data to fit the expected schema.
	Preparing and sending requests to the Ollama AI service API.
	Managing both streaming and non-streaming API responses.
	Processing API responses to convert them into suitable message formats.

The ChatOllamaAI struct has fields to configure the AI, including but not limited to:
	endpoint: URL of the Ollama AI service.
	model: The AI model used, e.g., "llama2:latest".
	receive_timeout: Max wait time for AI service responses.
	temperature: Influences the AI's response creativity.

For detailed info on on all other parameters see documentation here:
https://github.com/jmorganca/ollama/blob/main/docs/modelfile.md#valid-parameters-and-values
This module is for use within LangChain and follows the ChatModel behavior,
outlining callbacks AI chat models must implement.
Usage examples and more details are in the LangChain documentation or the
module's function docs.

Represents the OpenAI
ChatModel.
Parses and validates inputs for making a requests from the OpenAI Chat API.
Converts responses into more specialized LangChain data structures.
	https://github.com/openai/openai-cookbook/blob/main/examples/How_to_call_functions_with_chat_models.ipynb


  
    
  
  Callbacks


See the set of available callback: LangChain.ChatModels.LLMCallbacks

  
    
  
  Rate Limit API Response Headers


OpenAI returns rate limit information in the response headers. Those can be
accessed using the LLM callback on_llm_ratelimit_info like this:
handlers = %{
  on_llm_ratelimit_info: fn _model, headers ->
    IO.inspect(headers)
  end
}

{:ok, chat} = ChatOpenAI.new(%{callbacks: [handlers]})
When a request is received, something similar to the following will be output
to the console.
%{
  "x-ratelimit-limit-requests" => ["5000"],
  "x-ratelimit-limit-tokens" => ["160000"],
  "x-ratelimit-remaining-requests" => ["4999"],
  "x-ratelimit-remaining-tokens" => ["159973"],
  "x-ratelimit-reset-requests" => ["12ms"],
  "x-ratelimit-reset-tokens" => ["10ms"],
  "x-request-id" => ["req_1234"]
}

  
    
  
  Token Usage


OpenAI returns token usage information as part of the response body. That data
can be accessed using the LLM callback on_llm_token_usage like this:
handlers = %{
  on_llm_token_usage: fn _model, usage ->
    IO.inspect(usage)
  end
}

{:ok, chat} = ChatOpenAI.new(%{
  callbacks: [handlers],
  stream: true,
  stream_options: %{include_usage: true}
})
When a request is received, something similar to the following will be output
to the console.
%LangChain.TokenUsage{input: 15, output: 3}
The OpenAI documentation instructs to provide the stream_options with the
include_usage: true for the information to be provided.

  
    
  
  Tool Choice


OpenAI's ChatGPT API supports forcing a tool to be used.
	https://platform.openai.com/docs/api-reference/chat/create#chat-create-tool_choice

This is supported through the tool_choice options. It takes a plain Elixir map to provide the configuration.
By default, the LLM will choose a tool call if a tool is available and it determines it is needed. That's the "auto" mode.

  
    
  
  Example


For the LLM's response to make a tool call of the "get_weather" function.
ChatOpenAI.new(%{
  model: "...",
  tool_choice: %{"type" => "function", "function" => %{"name" => "get_weather"}}
})

  
    
  
  Azure OpenAI Support


To use ChatOpenAI with Microsoft's Azure hosted OpenAI models, the endpoint must be overridden and the API key needs to be provided in some way. The 

Parses and validates inputs for making a request for the Google AI  Chat API.
Converts response into more specialized LangChain data structures.

LangChain.Chains.DataExtractionChain 
    



      
Defines an LLMChain for performing data extraction from a body of text.
Provide the schema for desired information to be parsed into. It is treated as
though there are 0 to many instances of the data structure being described so
information is returned as an array.
Originally based on:
	https://github.com/langchain-ai/langchainjs/blob/main/langchain/src/chains/openai_functions/extraction.ts#L43


  
    
  
  Example


# JSONSchema definition of data we want to capture or extract.
schema_parameters = %{
  type: "object",
  properties: %{
    person_name: %{type: "string"},
    person_age: %{type: "number"},
    person_hair_color: %{type: "string"},
    dog_name: %{type: "string"},
    dog_breed: %{type: "string"}
  },
  required: []
}

# Model setup
{:ok, chat} = ChatOpenAI.new(%{temperature: 0})

# run the chain on the text information
data_prompt =
  "Alex is 5 feet tall. Claudia is 4 feet taller than Alex and jumps higher than him.
  Claudia is a brunette and Alex is blonde. Alex's dog Frosty is a labrador and likes to play hide and seek."

{:ok, result} = LangChain.Chains.DataExtractionChain.run(chat, schema_parameters, data_prompt)

# Example result
[
  %{
    "dog_breed" => "labrador",
    "dog_name" => "Frosty",
    "person_age" => nil,
    "person_hair_color" => "blonde",
    "person_name" => "Alex"
  },
  %{
    "dog_breed" => nil,
    "dog_name" => nil,
    "person_age" => nil,
    "person_hair_color" => "brunette",
    "person_name" => "Claudia"
  }
]
The schema_parameters in the previous example can also be expressed using a
list of LangChain.FunctionParam structs. An equivalent version looks like
this:
alias LangChain.FunctionParam

schema_parameters = [
  FunctionParam.new!(%{name: "person_name", type: :string}),
  FunctionParam.new!(%{name: "person_age", type: :number}),
  FunctionParam.new!(%{name: "person_hair_color", type: :string}),
  FunctionParam.new!(%{name: "dog_name", type: :string}),
  FunctionParam.new!(%{name: "dog_breed", type: :string})
]
|> FunctionParam.to_parameters_schema()

LangChain.Chains.SummarizeConversationChain 
    



      
When an AI conversation has many back-and-forth messages (from user to
assistant to user to assistant, etc.), the number of messages and the total
token count can be large. Large token counts present the following problems:
	Increased cost/price per generation
	Increased generation times
	Risk of exceeding the total token limit, resulting in an error

This chain is run as a separate process to summarize and condense a separate
conversation chain. It is assumed that the chain the user sees in the UI
retains all their original messages and they are not seeing the full, raw
message list.
We don't want to perform more work than we need to, so we'll only kick off the
summary process once the number of messages has reached some threshold, then
we'll retain a configured number of the most recent messages to help retain
continuity for the conversation.

  
    
  
  Options


	:llm - The LLM to use for performing the summarization. There is no need for streaming.
	:keep_count - The number of raw messages to retain. It will be the
most recent messages and defaults to 2 (a user and assistant message).
	:threshold_count - The total number of messages (excluding the system
message) that must be present before the summarizing operation is performed.
Running the summarization on a short conversation chain will return the
chain unchanged and not make any calls to an LLM.
	:override_system_prompt - When the system prompt should be customized for the instructions on how to summarize, this can be used to provide a customized replacement of the system prompt.
	:messages - When explicit control of multiple messages is needed, they can be provided as a list. They can be LangChain.PromptTemplates and the concatenated list of messages will be in the @conversation param. When this is used, any value in  :override_system_prompt is ignored.


  
    
  
  Examples


A basic example that processes the messages in a separate LLMChain, returning
an updated chain with summarized contents.
{:ok, summarized_chain} =
  %{
    llm: ChatOpenAI.new!(%{model: "gpt-4o-mini", stream: false}),
    keep_count: 2,
    threshold_count: 6
  }
  |> SummarizeConversationChain.new!()
  |> SummarizeConversationChain.summarize(chain_to_summarize)
Using a :with_fallback option to still try and summarize if the LLM errors from the Azure host OpenAI.
# Azure configured OpenAI LLM
fallback_llm =
  ChatOpenAI.new!(%{
    stream: false,
    endpoint: System.fetch_env!("AZURE_OPENAI_ENDPOINT"),
    api_key: System.fetch_env!("AZURE_OPENAI_KEY")
  })

{:ok, summarized_chain} =
  %{
    llm: ChatOpenAI.new!(%{model: "gpt-4o-mini", stream: false}),
    keep_count: 2,
    threshold_count: 6
  }
  |> SummarizeConversationChain.new!()
  |> SummarizeConversationChain.evaluate(chain_to_summarize, with_fallbacks: [fallback_llm])

LangChain.Chains.TextToTitleChain 
    



      
A convenience chain for turning a user's prompt text into a summarized title
for the anticipated conversation.

  
    
  
  Basic Examples


A basic example that generates a title
llm = ChatOpenAI.new!(%{model: "gpt-3.5-turbo", stream: false, seed: 0})
user_text = "Let's start a new blog post about the magical properties of pineapple cookies."

%{
  llm: llm,
  input_text: user_text
}
|> TextToTitleChain.new!()
|> TextToTitleChain.evaluate()

#=> "Magical Properties of Pineapple Cookies Blog Post"

  
    
  
  Examples using Title Examples


Want to get more consistent titles?
LLMs are pretty bad at following instructions for text length. However, we can
provide examples titles for the LLM to follow in format style and length. We
get the added benefit of getting more consistently formatted titles.
This is the same example, however now we provide other title examples to the
LLM to follow for consistency.
   llm = ChatOpenAI.new!(%{model: "gpt-3.5-turbo", stream: false, seed: 0})
user_text = "Let's start a new blog post about the magical properties of
pineapple cookies."

%{
  llm: llm,
  input_text: user_text,
  examples: [
    "Blog Post: Making Delicious and Healthy Smoothies",
    "System Email: Notifying Users of Planned Downtime"
  ]
}
|> TextToTitleChain.new!()
|> TextToTitleChain.evaluate()

#=> "Blog Post: Exploring the Magic of Pineapple Cookies"

  
    
  
  Overriding the System Prompt


For more explicit control of how titles are generated, an override_system_prompt can be provided.
%{
  llm: llm,
  input_text: user_text,
  override_system_prompt: ~s|
    You expertly summarize the User Text into a short 3 or 4 word title to represent a conversation in a positive way.|
}
|> TextToTitleChain.new!()
|> TextToTitleChain.evaluate()

  
    
  
  Using a Fallback


If the primary LLM fails to respond successfully, one or more fallback LLMs can be specified.
%{
  llm: primary_llm,
  input_text: user_text
}
|> TextToTitleChain.new!()
|> TextToTitleChain.evaluate(with_fallbacks: [fallback_llm])

      


      
        Summary


  
    Types
  


    
      
        
  
    
    LangChain.Message - LangChain v0.3.1
    
    

    


  
  

    
LangChain.Message 
    



      
Models a compl